BS2-+ Logger Made Simple

To help better understand the commands to and from the logger it would be useful to have a copy of the VDAP firmware commands close by while working through these examples

Logger connections

Pin 1 to VSS

Pin 2 not connected

Pin 3 to VDD

Pin 4 to Stamp P0

Pin 5 to Stamp P1

Pin 6 to VSS

Pin 7 not connected

Pin 8 not connected

Memory Stick Format

Before using any memory stick I advise that you format it. I also advise that you take control of the way the memory stick is formatted by using the following procedure (Win XP)

Control Panel- Administrative Tools  Computer Management - Disk Management 

Select the memory sticks drive letter and right click, select Format and format the drive as FAT 32 with an allocation unit size of 512

The following highlighted text can be copied and pasted into the Stamp editor 

The VDAP firmware is very specific about the commands the logger expects and the commands it returns. This means our code has to be very precise or we are going to lock the logger or corrupt the memory stick.

The BS2 is limited on the amount of serial data it can receive and handle; this situation can be eased if we run the data logger in Shortened Command Set. This means that the information returned from the logger is in an abbreviated form making it easier for the Stamp to deal with. 

If you take a look at the VDAP commands under the Response column you will see that most of the commands reply with <prompt>$0D, in shortened command set what we would see is ‘>’ CR. 

The arrow prompt (‘>’) plays a crucial part in our code, receiving the arrow is confirmation that the previously transmitted command was carried out successfully. If we don’t receive the arrow then we know there was an error and we can take action accordingly. 

The default baud rate for the Datalogger is 9600, I personally have found that the plain BS2 can be difficult to use with the logger at that rate even using the CTS/RTS flow control and that efforts to compensate for erroneous communications  will slow the logger to an unacceptable level. Therefore part of the initialization code in this document is to set communications to 4800 baud. Even at this rate it is possible to log in excess of 14 x 10 byte packets per second

Our program is going to consist of a Main section that will sequentially call 6 subroutines  

to report the file size in bytes, detect , open for writing , write , close and once more  report the file size in bytes

First we need our declarations we will start with the three constants that define the communications. These values can be changed to suit your design.

TX CON 0                     'P0 assigned TO SEROUT

RX CON 1                     'P1 assigned TO SERIN

Baud CON 188               'A baud rate of 4800

Next declare the variables.

GP_WORD VAR Word              'A general-purpose word variable

FILESIZE VAR Word             'Used TO measure file size

sData VAR Byte (12)       'A general-purpose array of 12 bytes

Filename VAR Byte(4)

counter VAR Word

idx VAR Word

Here is the Main section that calls the 6 sub routines

Main:

GOSUB DETECT

DEBUG "Enter a filename (Must be 4 characters)",CR

DEBUGIN STR Filename\4

DEBUG CR,CR,"Searching!!",CR

GOSUB SIZE

GOSUB OPEN_WRITE

GOSUB MEMORY_WRITE

GOSUB CLOSE

GOSUB SIZE

PAUSE 7000

GOTO main

Initialization and memory detection code is short and sweet although correct initialization is critical.  Most of the code for this part is DEBUG statements that provide us with instructions and inform us of what’s going on.  Lets work through the code for the subroutines.

Below is the code for initialization and detection and should be included with every logging program. The initial baud rate is 9600 (84) which is changed to 4800 (188) once initialization is established.

Step 1. Initialize and Detect

'***********[INITIALISE LOGGER & DETECT MEMORY STICK]*********

LOW TX

DEBUG HOME," Remove Memory and Press 1 ",CR

DO WHILE idx=0

GP_WORD=GP_WORD+1

DEBUGIN DEC1 idx

LOOP

DEBUG CLS

SEROUT TX, 84, ["IPA", CR]

PAUSE 200

SEROUT TX, 84, ["SCS", CR]

PAUSE 200

SEROUT TX,84, [$14,$20,$71,$02,$00,$0D]  'SET BAUD TO 4800

SERIN RX,baud,[WAIT(">")]

DEBUG "Initialized ",CR

PAUSE 1000

The DETECT sub routine provides a safe place to remove or insert a memory stick.

Return to this subroutine on completion of each task.

'****************[DETECT MEMORY STICK]*********************

DETECT:

  DEBUG CLS, HOME, "It is safe Insert or Remove Memory", CR,"(Remove and re-insert to go again)"

  SERIN RX, Baud,[WAIT ("DD")]

  DEBUG CLS, "Device Detected Please Wait...", CR

  PAUSE 3000

  DEBUG "Be patient.........",CR

  SERIN RX, Baud,[WAIT("No Up"),WAIT ("grade")]

  DEBUG "Logger is ready...", CR

  PAUSE 1500

RETURN

When a memory stick is inserted into the logger the logger goes through a sequence that must be completed before we can start to read or write data. First it sends back to the Stamp that a device was detected, “DD” in shortened command set, so our first SERIN waits for the characters “DD”. Secondly the logger examines the memory stick to see if it contains a firmware upgrade, when it has finished it reports back “No Upgrade” followed by a prompt “>” and that is the purpose of our second SERIN. If both of those steps are successful we can proceed setting the baud to 4800. 

Step 2. Open and Close

Our next two sub routines deal with opening and closing a file. 

The sub routine OPEN_WRITE will create and open a file if it does not exist, if it does exist the file will be opened and new data will be appended to the existing data.

'****************[OPEN FOR WRITE]*****************

OPEN_WRITE:

  'DEBUG CLS

  SEROUT TX, Baud,[$9,$20,STR Filename\4, ".txt", CR]

  SERIN RX, Baud,[WAIT (">")]

  DEBUG "File ",STR Filename\4,".txt open", CR

RETURN

Finally after we have finished with any open file we MUST close it.

'****************[CLOSE FILE]*********************

CLOSE:

  SEROUT TX, Baud,[$0A,$20,STR Filename\4 ,".txt", CR]

  SERIN RX, Baud,[WAIT (">")]

  DEBUG "File ",STR Filename\4,".txt closed", CR

RETURN

Before we can use these subroutines we need to be able to write some data. In the next section we will build a write sub routine.

Step 3. Write

The write routine is where you will probably have to give most thought. The write command has the number of bytes to write as a parameter, if you try and write more or less than the number you tell it to write the logger will lock up and possibly corrupt data on the memory stick. Depending on what you want to write or log governs the way you code the write sub routine. 

The first example writes the contents of the first 3 elements of array sData to three decimal places , thats 3 x 3 = 9 bytes so far. Our write string also contains the two separators “-” so thats another two bytes , 2 + 9 = 11 and finally we have a carriage return followed by a line feed (CR,$0A) to give us a total of 13 bytes ($0D) which you can see is the sixth HEX value of the write instruction.

The write routine repeats the above 20 times for a total of 260 bytes. Which file size will report after the first write.

'****************[MEMORY WRITE]*******************

MEMORY_WRITE:

  DEBUG "Writing ",CR

  FOR counter =0 TO 19

    GOSUB Get_Data:

    SEROUT TX, Baud, [$8,$20,$0,$0,$0,$0D,CR,DEC3 sData(0),"-",DEC3 sData(1),"-",DEC3 sData(2),CR,$0A,CR]

    SERIN RX, Baud,[WAIT (">")]

    DEBUG "."

  NEXT

  DEBUG CR

RETURN

Step 4. File size 

A simple routine to debug the file size in bytes to out monitor. 

NOTE: the variable Filesize is a word value , if the file is larger than 65535 bytes the size will not be read correctly

'****************[FILE SIZE]**********************

SIZE:

GP_WORD=0

SEROUT TX, Baud, [$01, $20,STR Filename\4,".txt", $0D]

SERIN RX, Baud, 5000, Not_Found, [WAIT($20),FILESIZE.LOWBYTE, FILESIZE.HIGHBYTE]

DEBUG  STR Filename\4," file size  ", DEC FILESIZE," Bytes", CR

RETURN

Not_Found:

GP_WORD=1

DEBUG CR, STR Filename\4,"  not found..."

RETURN

If the first SERIN times out then it is assumed the file does not exist and the program jumps to Not_Found:  and GP_VAR is set to 1 which flags to say the file does not exist. If the file does exist the GP_VAR returns with a value of 0 and allows us to continue. 

Notice the sub routine has two return statements, one for success and one for fail.

Finally here is the subroutine that generates the random numbers for our write routine.

'*********************************************

Get_Data:

FOR idx=0 TO 2

RANDOM GP_WORD

GP_WORD=GP_WORD//255

sData(idx)=GP_WORD

NEXT

RETURN



